This file accompanies the paper [LW] R. Laugwitz, G. Sanmarco: Finite-dimensional quantum groups of type Super A and non-semisimple modular categories, ArXiv preprint arXiv:2301.10685 and its eventual published version.

Author: Robert Laugwitz
Version of June 13, 2025.

Content: This file computes:

- The braiding, twist, evaluation etc. of the 4-dimensional simple object W = L(n+1,n) over u_q = u_q(sl_{2,I}), cf. [LW, Section 7.1].
- Several knot invariants obtained from this object, all knots with up to 7 crossings.
- The knot 10_132 which has the same Jones polynomial as 5_1 [D. Bar-Natan and S. Morrison, The Knot Atlas, 2022. Available at http://katlas.org, retrieved on
December 1, 2022.]
- The invariants of the links LL_2(1) and LL_2(2) of [S. Eliahou, L. H. Kauffman, and M. B. Thistlethwaite, Infinite families of links with trivial Jones polynomial, Topology 42 (2003), no. 1, 155–169.]

```
with(LinearAlgebra) : interface(rtablesize = 32) : with(linalg) :
with(ArrayTools) :
```

Basic setup/commands:

We define the evaluation and co-evuation (both for the left and right duals) for the 4-d module W = L(n+1,n).

```
ID4 := IdentityMatrix(4) : ID16 := IdentityMatrix(16) :
Signs4 := Matrix([[(−1)^n, 0, 0, 0], [0, 1, 0, 0], [0, 0, 1, 0], [0, 0, 0, (−1)^n]]);
```

$$Signs4 := \begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad \textbf{(1)}$$

```
ev4 := Matrix([1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1]) : evr4 := Multiply(Matrix([1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1]),
    KroneckerProduct(Signs4, ID4)) : coev4 := Matrix(⟨1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1, 0, 0, 0, 0, 1⟩) : coevr4 := Multiply(KroneckerProduct(Signs4,
    ID4), coev4) :
```

Tensoring with identities:

```
TensorID := proc(n :: integer, A :: Matrix, m :: integer) :: Matrix;
 description "IDn\otimes A\otimes \IDm";
   local B;
  B := KroneckerProduct(IdentityMatrix(n), KroneckerProduct(A, IdentityMatrix(m)));
 end proc;
```

```
TensorID := proc(n::integer, A::Matrix, m::integer)::Matrix;                  (2)
    local B;
    description "IDnotimes Aotimes IDm";
    B := LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(n), LinearAlgebra:-KroneckerProduct(A, LinearAlgebra:-
    IdentityMatrix(m)))
end proc
```

Tests: Dimension=4, qDimension=0.

```
Multiply(ev4, coev4);
Multiply(evr4, coev4);
```

$$\begin{bmatrix} 4 \end{bmatrix}$$

$$\begin{bmatrix} 0 \end{bmatrix} \quad \textbf{(3)}$$

Braiding H=\Psi_{W,W}: (n odd case)

We choose a large odd integer. This helps Maple to simplify powers (-1)^n.

```
n := 999;
```

$$n := 999 \quad \textbf{(4)}$$

Now we compute the R-matrix for W, W:

$$H := simplify\Big(Matrix\Big(\Big[\langle(-1)^n, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, (-1)^n\cdot(1+s^{-1}), 0, 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 2\cdot(-1)^n, 0, 0, 0, 0, 0, 1, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, (-1)^n\cdot(3+s^{-1}), 0, 0, 1+s^{-1}, 0, 0, 1+s^{-1}, 0, 0, (-1)^n, 0, 0, 0\rangle, \langle 0, -s^{-1}, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, (-1)^n\cdot s^{-1}, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, -2, 0, 0, 0, 0, 0, (-1)^{n+1}, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, (-1)^n\cdot(1+s^{-1}), 0, 0, 0, 0, 0, 1, 0, 0\rangle, \langle 0, 0, -1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, -2, 0, 0, (-1)^{n+1}, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, (-1)^n, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, (-1)^n\cdot 2, 0, 0, 1, 0\rangle, \langle 0, 0, 0, (-1)^n\cdot s^{-1}, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, -s^{-1}, 0, 0, 0, 0, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, -1, 0, 0, 0, 0\rangle, \langle 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, (-1)^n\rangle\Big]\Big)\Big);$$

$$H := \quad \mathbf{(5)}$$

$$\begin{bmatrix}
-1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & \frac{-s-1}{s} & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -3-\frac{1}{s} & 0 & 0 & -2 & 0 & 0 & -2 & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 \\
0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1+\frac{1}{s} & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{-s-1}{s} & 0 & 0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 \\
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 1+\frac{1}{s} & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2 & 0 & 0 & -1 & 0 \\
0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1
\end{bmatrix}$$

We compute the tensor products H12 = H\otimes Id_W and H23=Id_W \otimes H, followed by their inverses:

$H12 := KroneckerProduct(H, ID4) : \quad H23 := KroneckerProduct(ID4, H) :$

$Hinv := MatrixInverse(H) : Hinv12 := KroneckerProduct(Hinv, ID4) : Hinv23 := KroneckerProduct(ID4, Hinv) :$

We check that the twist of W is the identity:

$Signs4 := simplify(Signs4); evr4 := simplify(evr4) : TwistV := simplify(Multiply(KroneckerProduct(ID4, evr4), Multiply(H12, KroneckerProduct(ID4, coev4))));$

$$Signs4 := \begin{bmatrix} -1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 \end{bmatrix} \quad \mathbf{(6)}$$

$$TwistV := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \mathbf{(6)}$$

Now we define a **procedure to test a given matrix is zero**:

```
IsZeroMatrix :=proc(A :: Matrix)
local v, n, m, i, j;
 n, m := Size(A);
i := 1; j := 1;
 v := true;
 while (i < n and v = true) do
 while (j < m and v = true) do
 v := is(A[i, j] = 0);
 j := j + 1;
 end do;
 i := i + 1;
 end do;
 return v;
 end proc;
```

```
IsZeroMatrix := proc(A::Matrix)
    local v, n, m, i, j;
    n, m := ArrayTools:-Size(A);
    i := 1;
    j := 1;
    v := true;
    while i < n and v = true do while j < m and v = true do v := is(A[i, j] = 0); j := j + 1 end do; i := i + 1 end do;
    return v
end proc
```
(7)

```
IsZeroMatrix(H);
```

$$false \tag{8}$$

Test the QYBE:

```
IsZeroMatrix(simplify(Multiply(H23, Multiply(H12, H23)) − Multiply(H12, Multiply(H23, H12))));
```

$$true \tag{9}$$

The braiding with the dual L=\Psi_{W*,W}:

```
K := simplify(Multiply(TensorID(1, ev4, 16), Multiply(TensorID(4, Hinv, 4), TensorID(16, coev4, 1)))); Kinv := simplify(Multiply(TensorID(1,
    ev4, 16), Multiply(TensorID(4, H, 4), TensorID(16, coev4, 1)))) :
```

$$K := \begin{bmatrix} -1 & 0 & 0 & 0 & 0 & -s-1 & 0 & 0 & 0 & 0 & -2 & 0 & 0 & 0 & 0 & -s-3 \\ 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2\,s & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & -2\,s & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & \frac{s+1}{s} & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -s-1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -s & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & \frac{s+1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & -2 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 \end{bmatrix} \tag{10}$$

```
K12 := KroneckerProduct(K, ID4) : K23 := KroneckerProduct(ID4, K) : Kinv12 := KroneckerProduct(Kinv, ID4) :
Kinv23 := KroneckerProduct(ID4, Kinv) :
```

Test the QYBE:

$IsZeroMatrix(simplify(Multiply(K23, Multiply(K12, H23)) - Multiply(H12, Multiply(K23, K12))));$

$$true \tag{11}$$

The braiding with the dual K=\Psi_{W,W*}:

$L := MatrixInverse(Kinv); Linv := MatrixInverse(K):$

$$L := \begin{bmatrix} -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ s+1 & 0 & 0 & 0 & 0 & -s & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 2s & 0 & 0 & 0 & 0 & -s & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 2s & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\ 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -\frac{s+1}{s} & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & -\frac{s+1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 \\ -s-3 & 0 & 0 & 0 & 0 & s+1 & 0 & 0 & 0 & 0 & 2 & 0 & 0 & 0 & 0 & -1 \end{bmatrix} \tag{12}$$

$L12 := KroneckerProduct(L, ID4) : L23 := KroneckerProduct(ID4, L) : Linv12 := KroneckerProduct(Linv, ID4) :$
$Linv23 := KroneckerProduct(ID4, Linv) :$

Test the QYBE:

$IsZeroMatrix(simplify(Multiply(H23, Multiply(L12, L23)) - Multiply(L12, Multiply(L23, H12))));$

$$true \tag{13}$$

$IsZeroMatrix(simplify(Multiply(L23, Multiply(H12, K23)) - Multiply(K12, Multiply(H23, L12))));$

$$true \tag{14}$$

The dual braiding F=\Psi_{W*,W*}:

$F := simplify(Multiply(TensorID(1, ev4, 16), Multiply(TensorID(4, Kinv, 4), TensorID(16, coev4, 1)))); Finv := MatrixInverse(F):$

$$F := \left[\begin{array}{cccccccccccccccc}
-1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 \\
0 & 1 & 0 & 0 & \frac{-s-1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & -2 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\frac{1}{s} & 0 & 0 \\
0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & -2 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & -2 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1 & 0 \\
0 & 0 & 0 & -1 & 0 & 0 & 1+\frac{1}{s} & 0 & 0 & 1+\frac{1}{s} & 0 & 0 & -3-\frac{1}{s} & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & 0 & 0 & 0 & \frac{-s-1}{s} & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & 0 & -2 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -1
\end{array}\right] \quad \textbf{(15)}$$

$F12 := KroneckerProduct(F, ID4) : F23 := KroneckerProduct(ID4, F) : Finv12 := KroneckerProduct(Finv, ID4) :$
$Finv23 := KroneckerProduct(ID4, Finv) :$

Test the QYBE:

$IsZeroMatrix(simplify(Multiply(F23, Multiply(F12, F23)) - Multiply(F12, Multiply(F23, F12))));$

$$true \quad \textbf{(16)}$$

$IsZeroMatrix(simplify(Multiply(L23, Multiply(L12, F23)) - Multiply(F12, Multiply(L23, L12))));$

$$true \quad \textbf{(17)}$$

$IsZeroMatrix(simplify(Multiply(F23, Multiply(K12, K23)) - Multiply(K12, Multiply(K23, F12))));$

$$true \quad \textbf{(18)}$$

$IsZeroMatrix(simplify(Multiply(K23, Multiply(F12, L23)) - Multiply(L12, Multiply(F23, K12))));$

$$true \quad \textbf{(19)}$$

Computing invariants:

The knot 4_1:

$simplify(Multiply(TensorID(4, ev4, 1), Multiply(TensorID(1, Multiply(Kinv, Linv), 4), Multiply(TensorID(4, Multiply(L, K), 1), TensorID(1, coev4, 4)))));$

$$\left[\begin{array}{cccc}
\frac{6s^2+13s+6}{s} & 0 & 0 & 0 \\
0 & \frac{6s^2+13s+6}{s} & 0 & 0 \\
0 & 0 & \frac{6s^2+13s+6}{s} & 0 \\
0 & 0 & 0 & \frac{6s^2+13s+6}{s}
\end{array}\right] \quad \textbf{(20)}$$

Alternative tangle for 4_1:

simplify(*Multiply*(*TensorID*(1, *ev4*, 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*ev4*, *H*), 1), *Multiply*(*TensorID*(1, *KroneckerProduct*(*Finv*, *Hinv*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*coevr4*, *H*), 1), *TensorID*(1, *coevr4*, 4))))));

$$\begin{bmatrix} \frac{6s^2+13s+6}{s} & 0 & 0 & 0 \\ 0 & \frac{6s^2+13s+6}{s} & 0 & 0 \\ 0 & 0 & \frac{6s^2+13s+6}{s} & 0 \\ 0 & 0 & 0 & \frac{6s^2+13s+6}{s} \end{bmatrix} \tag{21}$$

The knot 5_2:

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1, *Multiply*(*Multiply*(*Kinv*, *Linv*), *Kinv*), 4), *Multiply*(*TensorID*(4, *Multiply*(*H*, *H*), 1), *TensorID*(1, *coevr4*, 4)))));

$$\begin{bmatrix} \frac{11s^3+22s^2+14s+2}{s^3} & 0 & 0 & 0 \\ 0 & \frac{11s^3+22s^2+14s+2}{s^3} & 0 & 0 \\ 0 & 0 & \frac{11s^3+22s^2+14s+2}{s^3} & 0 \\ 0 & 0 & 0 & \frac{11s^3+22s^2+14s+2}{s^3} \end{bmatrix} \tag{22}$$

The knot 6_1:

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1, *Multiply*(*Multiply*(*Multiply*(*Kinv*, *Linv*), *Kinv*), *Linv*), 4), *Multiply*(*TensorID*(4, *Multiply*(*L*, *K*), 1), *TensorID*(1, *coev4*, 4)))));

$$\begin{bmatrix} \frac{14s^3+35s^2+26s+6}{s^2} & 0 & 0 & 0 \\ 0 & \frac{14s^3+35s^2+26s+6}{s^2} & 0 & 0 \\ 0 & 0 & \frac{14s^3+35s^2+26s+6}{s^2} & 0 \\ 0 & 0 & 0 & \frac{14s^3+35s^2+26s+6}{s^2} \end{bmatrix} \tag{23}$$

The knot 6_2:

simplify(*Multiply*(*TensorID*(4, *evr4*, 1), *Multiply*(*TensorID*(1, *MatrixPower*(*H*, 3), 4), *Multiply*(*TensorID*(4, *Kinv*, 1), *Multiply*(*TensorID*(1, *K*, 4), *Multiply*(*TensorID*(4, *Hinv*, 1), *TensorID*(1, *coevr4*, 4)))))));

$$\begin{bmatrix} \frac{14 s^4 + 39 s^3 + 40 s^2 + 22 s + 6}{s^3} & 0 & 0 & \cdots \\ 0 & \frac{14 s^4 + 39 s^3 + 40 s^2 + 22 s + 6}{s^3} & 0 & \cdots \\ 0 & 0 & \frac{14 s^4 + 39 s^3 + 40 s^2 + 22 s + 6}{s^3} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(24)}$$

The knot 6_3:

simplify(*Multiply*(*TensorID*(1, *ev4*, 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*ev4*, *H*), 1), *Multiply*(*KroneckerProduct*(*MatrixPower*(*Finv*, 2), *TensorID*(1, *Hinv*, 4)), *Multiply*(*TensorID*(4, *KroneckerProduct*(*coevr4*, *Multiply*(*H*, *H*)), 1), *TensorID*(1, *coevr4*, 4))))));

$$\begin{bmatrix} \frac{10 s^4 + 42 s^3 + 65 s^2 + 42 s + 10}{s^2} & 0 & 0 & \cdots \\ 0 & \frac{10 s^4 + 42 s^3 + 65 s^2 + 42 s + 10}{s^2} & 0 & \cdots \\ 0 & 0 & \frac{10 s^4 + 42 s^3 + 65 s^2 + 42 s + 10}{s^2} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(25)}$$

The knot 7_2:

simplify(*Multiply*(*TensorID*(1, *ev4*, 4), *Multiply*(*TensorID*(4, *Multiply*(*Hinv*, *Hinv*), 1), *Multiply*(*TensorID*(1, *Multiply*(*Multiply*(*Multiply*(*Multiply*(*L*, *K*), *L*), *K*), *L*), 4), *TensorID*(4, *coev4*, 1)))));

$$\begin{bmatrix} -8 s^4 - 40 s^3 - 72 s^2 - 74 s - 31 & 0 & 0 & \cdots \\ 0 & 2 s^4 + 14 s^3 + 40 s^2 + 38 s + 11 & 0 & \cdots \\ 0 & 0 & -2 s^4 + 2 s^3 + 40 s^2 + 46 s + 19 & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(26)}$$

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1, *Multiply*(*Multiply*(*Multiply*(*Multiply*(*Kinv*, *Linv*), *Kinv*), *Linv*), *Kinv*), 4), *Multiply*(*TensorID*(4, *Multiply*(*H*, *H*), 1), *TensorID*(1, *coevr4*, 4)))));

$$\begin{bmatrix} \frac{19 s^4 + 46 s^3 + 40 s^2 + 14 s + 2}{s^4} & 0 & 0 & \cdots \\ 0 & \frac{19 s^4 + 46 s^3 + 40 s^2 + 14 s + 2}{s^4} & 0 & \cdots \\ 0 & 0 & \frac{19 s^4 + 46 s^3 + 40 s^2 + 14 s + 2}{s^4} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(27)}$$

The knot 7_3:

simplify(*Multiply*(*TensorID*(4, *evr4*, 1), *Multiply*(*TensorID*(1, *MatrixPower*(*Hinv*, 4), 4), *Multiply*(*TensorID*(4, *Multiply*(*Multiply*(*K*, *L*), *K*), 1), *TensorID*(1, *coev4*, 4)))));

$$\begin{bmatrix} 2s^5+14s^4+32s^3+50s^2+50s+21 & 0 & 0 & \cdots \\ 0 & 2s^5+14s^4+32s^3+50s^2+50s+21 & 0 & \cdots \\ 0 & 0 & 2s^5+14s^4+32s^3+50 & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(28)}$$

The knot 7_4:

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1 , *KroneckerProduct*(*ev4*, *Multiply*(*K*, *L*)), 4), *Multiply*(*TensorID*(4, *Hinv*, 4·4), *Multiply*(*TensorID*(1, *L*, 4·4·4), *Multiply*(*TensorID*(16, *K*, 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*Finv*, *Hinv*), 1), *TensorID*(1, *KroneckerProduct*(*coev4*, *coevr4*), 4)))))))));

$$\begin{bmatrix} 2s^4+24s^3+76s^2+88s+35 & 0 & 0 & \cdots \\ 0 & 2s^4+24s^3+76s^2+88s+35 & 0 & \cdots \\ 0 & 0 & 2s^4+24s^3+76s^2+88s+35 & \cdots \\ 0 & 0 & 0 & 2s^{\cdot}\cdots \end{bmatrix} \quad \textbf{(29)}$$

The knot 7_5:

simplify(*Multiply*(*TensorID*(4, *evr4*, 1), *Multiply*(*TensorID*(1, *MatrixPower*(*H*, 3), 4), *Multiply*(*TensorID*(4, *Multiply*(*Kinv*, *Linv*), 1), *Multiply*(*TensorID*(1, *H*, 4), *Multiply*(*TensorID*(4, *Kinv*, 1), *TensorID*(1, *coev4*, 4)))))));

$$\begin{bmatrix} \frac{29s^5+82s^4+96s^3+60s^2+20s+2}{s^5} & 0 & 0\ \cdots \\ 0 & \frac{29s^5+82s^4+96s^3+60s^2+20s+2}{s^5} & 0\ \cdots \\ 0 & 0 & \frac{29s^5+82s^4+96s^3}{s^5}\cdots \\ 0 & 0 & 0\ \cdots \end{bmatrix} \quad \textbf{(30)}$$

The knot 7_6:

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1, *KroneckerProduct*(*evr4*, *Multiply*(*K*, *L*)), 4), *Multiply*(*TensorID*(4, *Linv*, 4·4), *Multiply*(*TensorID*(1 , *KroneckerProduct*(*H*, *F*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*coev4*, *Multiply*(*Linv*, *Kinv*)), 1), *TensorID*(1, *coev4*, 4)))))));

$$\begin{bmatrix} \frac{30s^4+105s^3+134s^2+76s+16}{s^3} & 0 & 0 & \cdots \\ 0 & \frac{30s^4+105s^3+134s^2+76s+16}{s^3} & 0 & \cdots \\ 0 & 0 & \frac{30s^4+105s^3+134s^2+76s+}{s^3} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad \textbf{(31)}$$

The knot 7_7:

simplify(*Multiply*(*TensorID*(4, *ev4*, 1), *Multiply*(*TensorID*(1, *Multiply*(*K*, *L*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*ev4*, *Linv*), 1), *Multiply*(*TensorID*(1 , *KroneckerProduct*(*Multiply*(*K*, *L*), *H*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*coevr4*, *Kinv*), 1), *TensorID*(1, *coev4*, 4)))))));

$$\begin{bmatrix} \frac{32 s^4+124 s^3+171 s^2+96 s+18}{s^2} & 0 & 0 & \cdots \\ 0 & \frac{32 s^4+124 s^3+171 s^2+96 s+18}{s^2} & 0 & \cdots \\ 0 & 0 & \frac{32 s^4+124 s^3+171 s^2+96 s+}{s^2} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad (32)$$

The knot 10_132 has the same Jones Polyonomial as 5_1. Here we compute its invariant \rI_W:

simplify(*Multiply*(*TensorID*(1, *ev4*, 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*ev4*, *H*), 1), *Multiply*(*TensorID*(1, *KroneckerProduct*(*Finv*, *H*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*Linv*, *KroneckerProduct*(*Hinv*, *ev4*)), 1), *Multiply*(*TensorID*(1, *KroneckerProduct*(*KroneckerProduct*(*Linv*, *L*), *Kinv*), 4), *Multiply*(*TensorID*(4, *KroneckerProduct*(*KroneckerProduct*(*coevr4*, *F*), *H*), 1), *TensorID*(1, *KroneckerProduct*(*coev4*, *coevr4*), 4)))))))

$$\begin{bmatrix} \frac{4 s^6+4 s^5-3 s^4+10 s^2+8 s+2}{s^4} & 0 & 0 & \cdots \\ 0 & \frac{4 s^6+4 s^5-3 s^4+10 s^2+8 s+2}{s^4} & 0 & \cdots \\ 0 & 0 & \frac{4 s^6+4 s^5-3 s^4+10 s^2+8 s+2}{s^4} & \cdots \\ 0 & 0 & 0 & \cdots \end{bmatrix} \quad (33)$$

Computing link invariants of LL2(1) and LL2(2):

Some general procedures

```
coevs := proc(n :: integer) :: Matrix;
description "produces a coevaluation matrix concatenated n times";
  local i, j, A;
 if n = 1 then A := coev4; else
   A := coev4; j := 4;
   for i from 2 to n do
    A := Multiply(KroneckerProduct(KroneckerProduct(IdentityMatrix(j), coev4), IdentityMatrix(j)), A);
j := j·4;
 end do;
 A;
end if;
end proc;
```

```
coevs := proc(n::integer)::Matrix;                                                    (34)
    local i, j, A;
    description "produces a coevaluation matrix concatenated n times";
    if n = 1 then
        A := coev4
    else
        A := coev4;
        j := 4;
        for i from 2 to n do
            A := Multiply(LinearAlgebra:-KroneckerProduct(LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(j), coev4),
            LinearAlgebra:-IdentityMatrix(j)), A);
            j := 4*j
        end do;
        A
```

```
    end if
end proc
 evrs := proc(n :: integer) :: Matrix;
 description "produces a right evaluation matrix concatenated n times";
     local i, j, A;
    if n = 1 then A := evr4; else
      A := evr4; j := 4;
      for i from 2 to n do
       A := Multiply(A, KroneckerProduct(KroneckerProduct(IdentityMatrix(j), evr4), IdentityMatrix(j)));
j := j·4;
    end do;
   A;
 end if;
 end proc;
```

```
evrs := proc(n::integer)::Matrix;                                                                  (35)
    local i, j, A;
    description "produces a right evaluation matrix concatenated n times";
    if n = 1 then
        A := evr4
    else
        A := evr4;
        j := 4;
        for i from 2 to n do
            A := Multiply(A, LinearAlgebra:-KroneckerProduct(LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(j), evr4),
            LinearAlgebra:-IdentityMatrix(j)));
            j := 4*j
        end do;
        A
    end if
end proc
```

```
 evs := proc(n :: integer) :: Matrix;
 description "produces a left evaluation matrix concatenated n times";
     local i, j, A;
    if n = 1 then A := ev4; else
      A := ev4; j := 4;
      for i from 2 to n do
       A := Multiply(A, KroneckerProduct(KroneckerProduct(IdentityMatrix(j), ev4), IdentityMatrix(j)));
j := j·4;
    end do;
   A;
 end if;
 end proc;
```

```
evs := proc(n::integer)::Matrix;                                                                   (36)
    local i, j, A;
    description "produces a left evaluation matrix concatenated n times";
    if n = 1 then
        A := ev4
    else
        A := ev4;
        j := 4;
        for i from 2 to n do
            A := Multiply(A, LinearAlgebra:-KroneckerProduct(LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(j), ev4),
            LinearAlgebra:-IdentityMatrix(j)));
            j := 4*j
        end do;
        A
    end if
end proc
```

```
Braidat := proc(n :: integer, i :: integer) :: Matrix;
 description "braids the i and i+1-th copies in an n-fold tensor product";
    local B;
     B := KroneckerProduct(IdentityMatrix(4^(i-1)), KroneckerProduct(H, IdentityMatrix(4^(n-i-1))));
 end proc;
Braidinvat := proc(n :: integer, i :: integer) :: Matrix;
 description "braids the i and i+1-th copies in an n-fold tensor product";
    local B;
     B := KroneckerProduct(IdentityMatrix(4^(i-1)), KroneckerProduct(Hinv, IdentityMatrix(4^(n-i-1))));
 end proc;
```

```
Braidat := proc(n::integer, i::integer)::Matrix;
    local B;
    description "braids the i and i+1-th copies in an n-fold tensor product";
    B := LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(4^(i - 1)), LinearAlgebra:-KroneckerProduct(H, LinearAlgebra:-
    IdentityMatrix(4^(n - i - 1))))
end proc
```

```
Braidinvat := proc(n::integer, i::integer)::Matrix;                                        (37)
    local B;
    description "braids the i and i+1-th copies in an n-fold tensor product";
    B := LinearAlgebra:-KroneckerProduct(LinearAlgebra:-IdentityMatrix(4^(i - 1)), LinearAlgebra:-KroneckerProduct(Hinv,
    LinearAlgebra:-IdentityMatrix(4^(n - i - 1))))
end proc
```

```
Braid := proc(n :: integer, m :: integer) :: Matrix;
 description "braids n strands past m strands";
    local i, j, B;
     B := IdentityMatrix(4^(n+m));
     for j from 1 to m do
 for i from 0 to n - 1 do
      B := Multiply(Braidat(n + m, n - i + j - 1), B);
   end do;
 end do;
   B;
 end proc;
 Braid(2, 1) :
```

```
Braid := proc(n::integer, m::integer)::Matrix;                                             (38)
    local i, j, B;
    description "braids n strands past m strands";
    B := LinearAlgebra:-IdentityMatrix(4^(n + m));
    for j to m do for i from 0 to n - 1 do B := Multiply(Braidat(n + m, n - i + j - 1), B) end do end do;
    B
end proc
```

```
Braidinv := proc(n :: integer, m :: integer) :: Matrix;
 description "braids n strands past m strands";
    local i, j, B;
     B := IdentityMatrix(4^(n+m));
     for j from 1 to m do
 for i from 0 to n - 1 do
      B := Multiply(Braidinvat(n + m, n - i + j - 1), B);
   end do;
 end do;
   B;
 end proc;
```

```
Braidinv := proc(n::integer, m::integer)::Matrix;                                          (39)
    local i, j, B;
    description "braids n strands past m strands";
    B := LinearAlgebra:-IdentityMatrix(4^(n + m));
    for j to m do for i from 0 to n - 1 do B := Multiply(Braidinvat(n + m, n - i + j - 1), B) end do end do;
    B
end proc
```

Now we can implement the **link LL_2(1):**

$LL21 := simplify(Multiply(KroneckerProduct(ID4, evrs(3)), Multiply(KroneckerProduct(Multiply(MatrixPower(Braidinv(2, 2), 2), KroneckerProduct(MatrixPower(Hinv, 3), MatrixPower(H, 3))), IdentityMatrix(4 \cdot 4 \cdot 4)), KroneckerProduct(ID4, coevs(3)))));$

$$LL21 := \left[\begin{array}{ccc} \frac{2 s^9 + 4 s^8 - 2 s^7 - 6 s^6 - 4 s^5 + 4 s^3 + 4 s^2 - 2}{s^2} & 0 & \cdots \\ 0 & \frac{2 s^9 + 4 s^8 - 2 s^7 - 6 s^6 - 4 s^5 + 4 s^3 + 4 s^2 - 2}{s^2} & \cdots \\ 0 & 0 & \frac{2}{}\cdots \\ 0 & 0 & \cdots \end{array}\right] \tag{40}$$

Computing I_W(LL_2(2)): We define two morphisms Beta1 and Beta2 separately:

$Beta1 := simplify(Multiply(KroneckerProduct(Multiply(Kinv, Linv), ID4), Multiply(KroneckerProduct(ID4, L), Multiply(KroneckerProduct(MatrixPower(Hinv, 5), ID4), KroneckerProduct(ID4, coev4))))):$

$Beta2pre := simplify(Multiply(KroneckerProduct(ID4, KroneckerProduct(Linv, ID4)), Multiply(KroneckerProduct(Hinv, Linv), Multiply(KroneckerProduct(KroneckerProduct(ID4, MatrixPower(Hinv, 2)), ID4), Multiply(KroneckerProduct(Hinv, Kinv), KroneckerProduct(ID4, KroneckerProduct(Kinv, ID4))))))):$

$Beta2 := simplify(Multiply(KroneckerProduct(Beta2pre, ID16), KroneckerProduct(ID16, coevs(2)))):$

$Beta3 := simplify(Multiply(KroneckerProduct(ID4, ev4), Multiply(KroneckerProduct(Multiply(K, L), ID4), Multiply(KroneckerProduct(ID4, Linv), KroneckerProduct(MatrixPower(H, 5), ID4))))):$

$Beta4 := simplify(Multiply(KroneckerProduct(ID4, Multiply(ev4, KroneckerProduct(KroneckerProduct(ID4, evr4), ID4))), KroneckerProduct(KroneckerProduct(ID16, Beta3), ID16))):$

$LL22 := simplify(Multiply(Beta4, Multiply(KroneckerProduct(Beta2, ID4), Beta1)));$

$$LL22 := \left[\begin{array}{c} \frac{2\left(8 s^{16} + 30 s^{15} + 49 s^{14} + 62 s^{13} + 84 s^{12} + 77 s^{11} - 18 s^{10} - 105 s^9 - 99 s^8 - 150 s^7 - 213 s^6 - 113 s^5 + 40 s^4 + 1\right.}{s^8}\cdots \\ 0 \quad \cdots \\ 0 \quad \cdots \\ 0 \quad \cdots \end{array}\right] \tag{41}$$